

# Vers la programmation logique

SAMUEL GALLAY

2 mars 2020

## 1 L'écriture en OCaml d'un interpréteur du langage Prolog

Prolog, pour programmation en logique est un langage créé par A. Colmerauer, en 1972 à Marseille. La programmation logique est un paradigme différent de la programmation impérative ou de la programmation fonctionnelle. On définit des faits élémentaires et des règles de la manière suivante :

```
apprend(eve, mathematiques).
apprend(benjamin, informatique).
apprend(benjamin, physique).
enseigne(alice, physique).
enseigne(pierre, mathematiques).
enseigne(pierre, informatique).
```

```
etudiant_de(E,P):-apprend(E,M), enseigne(P,M).
```

Ces informations décrivent l'ensemble des connaissances du programme. On lit ici “Benjamin apprend l'informatique”, ou “Alice enseigne la physique”. La dernière ligne est une règle : “L'élève E est étudiant du professeur P si E apprend la matière M et que P enseigne M”. Le symbole `-` se lit *si*, et la virgule entre `apprend(E,M)` et `enseigne(P,M)` signifie *et*. Ensuite l'utilisateur peut effectuer des requêtes comme ceci :

```
?-etudiant_de(E, pierre)
```

L'interpréteur Prolog renvoie alors `E = benjamin` et `E = eve`.

Ce qui différencie la programmation logique des formes plus courantes d'informatique impérative, c'est qu'on n'explique pas au programme comment résoudre la requête, on se contente de décrire le problème.

De très nombreuses introductions au langage Prolog peuvent être trouvées sur internet, je n'en cite qu'une seule qui m'est apparue comme convenable pour une première approche : *Simply Logical : Intelligent Reasoning by Example*, P. Flach, 1994 [2].

### 1.1 La représentation des programmes Prolog

La représentation des programmes Prolog dans OCaml se fait de la manière suivante :

- `var` est une variable (commençant par une majuscule en Prolog), identifiée par une chaîne de caractères, et un numéro. Le numéro sera utilisé pour renommer facilement les variables, pour éviter des conflits de noms.

- `atom` est une chaîne de caractères représentant les symboles de prédicats : soit des constantes (arité 0 du prédicat), soit les foncteurs (arité strictement positive). L'arité de `etudiant_de` est 2.
- un `term` est soit une variable, soit sous la forme `prédicat(terme1, terme2, ...)`, qui est donc un atome, et une liste de termes.
- une `clause` est de la forme `etudiant_de(E,P):-apprend(E,M), enseigne(P,M).`, un terme à gauche et une liste de termes à droite. `eleve(samuel).` est une clause, il n'y rien à droite.

```
[1]: type var = Id of string * int;;
type atom = Atom of string;;
type term = Var of var | Term of atom * (term list);;
type clause = Clause of term * (term list);;
```

```
[1]: type var = Id of string * int
```

```
[1]: type atom = Atom of string
```

```
[1]: type term = Var of var | Term of atom * term list
```

```
[1]: type clause = Clause of term * term list
```

Ce n'est pas moi qui ai inventé cette structure particulière : je l'ai retrouvée dans un diapo de cours en Haskell sur le langage Prolog écrit par Alan Smaill en 2009 [6].

Une structure proche est utilisée en Lisp par Peter Norvig dans *Paradigms of Artificial Intelligence Programming*, 1992 [4]. Une citation qui permet de comprendre la structure d'un programme en Prolog : "We will build a single uniform data base of clauses, without distinguishing rules from facts. The simplest representation of clauses is as a cons cell holding the head and the body. For facts, the body will be empty."

J'explique plus bas le *pourquoi* de cette structure, qui permet de représenter les clauses de Horn.

## 1.2 L'analyse lexicale : le lexer

Il faut d'abord transformer le programme qui est une chaîne de caractères, en une liste de symboles. Les `token` sont les différents symboles présents dans un programme Prolog simple. Des variables et des atomes, et ces caractères spéciaux : `( | ) | , | . | :-`.

```
[2]: type token = TLPare | TRPare | TDot | TIif | TAnd | TVar of string | TAtom of  
↳ string;;
```

```
[2]: type token =
      TLPare
    | TRPare
    | TDot
```

```

| TIf
| TAnd
| TVar of string
| TAtom of string

```

```

[3]: (* Transforme une chaîne de caractères en liste et réciproquement *)
let to_list str = List.init (String.length str) (String.get str);;
let to_string lst = String.init (List.length lst) (List.nth lst);;

(* Enlève les espaces, tabulations de la liste de caractères *)
let rec remove_spaces = function
| (' ' | '\t' | '\n')::l -> remove_spaces l
| l -> l;;

(* Retire le premier mot (suite de lettres et de '_' ) de la liste *)
let rec read_word char_list mot = match char_list with
| ('A'..'Z' | 'a'..'z' | '_' as c)::l -> read_word l (c::mot)
| char_list -> char_list, to_string (List.rev mot);;

(* Pour séparer les variables et les atomes *)
let string_to_atomvar = function
| str -> (match String.get str 0 with
          | ('A'..'Z') -> TVar str
          | ('a'..'z') -> TAtom str
          | _ -> failwith "Wrong Word")

(* Transforme la liste de caractères en liste de symboles *)
let rec charlist_to_tokenlist liste_car = let lst = remove_spaces liste_car in
  match lst with
  | [] -> []
  | '.'::l -> TDot::(charlist_to_tokenlist l)
  | '('::l -> TLPare::(charlist_to_tokenlist l)
  | ')'::l -> TRPare::(charlist_to_tokenlist l)
  | ':'::l -> TIf::(charlist_to_tokenlist l)
  | ','::l -> TAnd::(charlist_to_tokenlist l)
  | ('A'..'Z' | 'a'..'z' | '_' )::l -> let l, m = read_word lst []
  in (string_to_atomvar m)::(charlist_to_tokenlist l)
  | l -> failwith ("Wrong Syntax in " ^ (to_string l));;

(* Transforme la chaîne de caractères en liste de symboles *)
let lexer str = charlist_to_tokenlist (to_list str);;

```

```

[3]: val to_list : string -> char list = <fun>

```

```
[3]: val to_string : char list -> string = <fun>
```

```
[3]: val remove_spaces : char list -> char list = <fun>
```

```
[3]: val read_word : char list -> char list -> char list * string = <fun>
```

```
[3]: val string_to_atomvar : string -> token = <fun>
```

```
[3]: val charlist_to_tokenlist : char list -> token list = <fun>
```

```
[3]: val lexer : string -> token list = <fun>
```

### 1.2.1 Un exemple d'utilisation de l'analyseur lexical :

```
[4]: let program = "apprend(eve, mathematiques).  
apprend(benjamin, informatique).  
apprend(benjamin, physique).  
enseigne(alice, physique).  
enseigne(pierre, mathematiques).  
enseigne(pierre, informatique).  
etudiant_de(E,P):-apprend(E,M), enseigne(P,M).  
egal(R,R).";;  
  
lexer program;;
```

```
[4]: val program : string =  
    "apprend(eve, mathematiques).\napprend(benjamin,  
informatique).\napprend(benjamin, physique).\nenseigne(alice,  
physique).\nenseigne(pierre, mathematiques).\nenseigne(pierre,  
informatique).\netudiant_de(E,P):-apprend(E,M), enseigne(P,M).\negal(R,R)."
```

```
[4]: - : token list =  
[TAtom "apprend"; TLPAr; TAtom "eve"; TAnd; TAtom "mathematiques"; TRPar;  
TDot; TAtom "apprend"; TLPAr; TAtom "benjamin"; TAnd; TAtom "informatique";  
TRPar; TDot; TAtom "apprend"; TLPAr; TAtom "benjamin"; TAnd;  
TAtom "physique"; TRPar; TDot; TAtom "enseigne"; TLPAr; TAtom "alice"; TAnd;  
TAtom "physique"; TRPar; TDot; TAtom "enseigne"; TLPAr; TAtom "pierre";  
TAnd; TAtom "mathematiques"; TRPar; TDot; TAtom "enseigne"; TLPAr;  
TAtom "pierre"; TAnd; TAtom "informatique"; TRPar; TDot;  
TAtom "etudiant_de"; TLPAr; TVar "E"; TAnd; TVar "P"; TRPar; TIf;
```

```
TAtom "apprend"; TLPAr; TVar "E"; TAnd; TVar "M"; TRPar; TAnd;
TAtom "enseigne"; TLPAr; TVar "P"; TAnd; TVar "M"; TRPar; TDot;
TAtom "egal"; TLPAr; TVar "R"; TAnd; TVar "R"; TRPar; TDot]
```

Quelques remarques sur le lexer : j'ai bien repris les idées du livre *Le Langage Caml*, Pierre Weis et Xavier Leroy, 1992 [7]. Une grande différence est l'utilisation des List à la place des Stream. L'utilisation des Stream semble plus intelligente, mais ils ont pour des raisons qui m'échappent ils ont presque disparu des dernières versions d'OCaml (le pattern-matching est beaucoup plus difficile). Avec les listes il y a bien plus de copies, mais nous n'avons pas là de véritable problème de performance.

### 1.3 L'analyse syntaxique : le parser

Il faut maintenant transformer la liste de tokens en une structure formée des types `clause`, `term`, `atom` et `var`.

#### 1.3.1 Isoler les clauses séparées par des points

```
[5]: (* Sépare les différentes clauses *)
let extract_clauses token_list =
let rec f tk_lst tmp = match tk_lst with
| [] -> if tmp = [] then [] else failwith "No end of clause"
| TDot::l -> (List.rev tmp)::(f l [])
| t::l -> f l (t::tmp)
in f token_list [];

extract_clauses (lexer program);;
```

```
[5]: val extract_clauses : token list -> token list list = <fun>
```

```
[5]: - : token list list =
[[TAtom "apprend"; TLPAr; TAtom "eve"; TAnd; TAtom "mathematiques"; TRPar];
 [TAtom "apprend"; TLPAr; TAtom "benjamin"; TAnd; TAtom "informatique";
  TRPar];
 [TAtom "apprend"; TLPAr; TAtom "benjamin"; TAnd; TAtom "physique"; TRPar];
 [TAtom "enseigne"; TLPAr; TAtom "alice"; TAnd; TAtom "physique"; TRPar];
 [TAtom "enseigne"; TLPAr; TAtom "pierre"; TAnd; TAtom "mathematiques";
  TRPar];
 [TAtom "enseigne"; TLPAr; TAtom "pierre"; TAnd; TAtom "informatique"; TRPar];
 [TAtom "etudiant_de"; TLPAr; TVar "E"; TAnd; TVar "P"; TRPar; TIf;
  TAtom "apprend"; TLPAr; TVar "E"; TAnd; TVar "M"; TRPar; TAnd;
  TAtom "enseigne"; TLPAr; TVar "P"; TAnd; TVar "M"; TRPar];
 [TAtom "egal"; TLPAr; TVar "R"; TAnd; TVar "R"; TRPar]]
```

### 1.3.2 Isoler les termes de droite et de gauche d'une clause séparés par :-

```
[6]: (* Sépare les membres de droite et de gauche d'une clause *)
let extract_rightleft cl =
  let rec f cl left = match cl with
  | [] -> (List.rev left), []
  | TIf::l -> (List.rev left), l
  | t::l -> f l (t::left)
  in f cl [];

extract_rightleft (lexer "etudiant_de(E,P):-apprend(E,M), enseigne(P,M)");;
```

```
[6]: val extract_rightleft : token list -> token list * token list = <fun>
```

```
[6]: - : token list * token list =
([TAtom "etudiant_de"; TLPar; TVar "E"; TAnd; TVar "P"; TRPar],
 [TAtom "apprend"; TLPar; TVar "E"; TAnd; TVar "M"; TRPar; TAnd;
  TAtom "enseigne"; TLPar; TVar "P"; TAnd; TVar "M"; TRPar])
```

### 1.3.3 Isoler les différents termes d'une suite de termes séparés par des virgules

```
[7]: (* Sépare les termes d'une liste de termes *)
let extract_term_list tk_lst =
let rec f tk_lst term_tmp lvl = match tk_lst with
  | [] -> [term_tmp]
  | TLPar::l -> f l (term_tmp @ [TLPar]) (lvl + 1)
  | TRPar::l -> if lvl > 0 then f l (term_tmp @ [TRPar]) (lvl - 1)
                 else failwith "Mismatched Parenthesis"
  | TAnd::l when lvl = 0 -> term_tmp::(f l [] 0)
  | t::l -> f l (term_tmp @ [t]) lvl
in f tk_lst [] 0;;

extract_term_list (lexer "apprend(E,M), enseigne(P,M)");;
```

```
[7]: val extract_term_list : token list -> token list list = <fun>
```

```
[7]: - : token list list =
[[TAtom "apprend"; TLPar; TVar "E"; TAnd; TVar "M"; TRPar];
 [TAtom "enseigne"; TLPar; TVar "P"; TAnd; TVar "M"; TRPar]]
```

### 1.3.4 Transformer la liste de symboles en un terme

```
[8]: (* Renvoie l'intérieur d'une liste de symboles dont les extrêmes sont des ↵
      ↪ parenthèses *)
let inside_parenthesis = function
| [] -> failwith "Empty List"
| TLPPar::l -> (let rec f = function
                | [] -> failwith "Mismatched Parenthesis"
                | [TRPar] -> []
                | h::t -> h::(f t)
                in f l)
| _ -> failwith "List is not beginning with TLPPar";;

inside_parenthesis (lexer "(E,M)")
```

```
[8]: val inside_parenthesis : token list -> token list = <fun>
```

```
[8]: - : token list = [TVar "E"; TAnd; TVar "M"]
```

```
[9]: (* Prend une liste de tokens qui est censée ne représenter qu'un terme ↵
      ↪ 'mouton', 'élève(samuel)', 'X',
      ↪ 'apprend(X, Y)' et renvoie ce terme *)
let rec tokenlist_to_term = function
| [] -> failwith "Error : [] is not a term"
| (TVar str)::[] -> Var (Id (str, 0))
| (TVar str)::_ -> failwith "No other arguments with Var"
| (TAtom str)::[] -> Term (Atom str, [])
| (TAtom str)::l -> Term (Atom str, List.map tokenlist_to_term ↵
      ↪ (extract_term_list (inside_parenthesis l)))
| _ -> failwith "Not a valid term";;

tokenlist_to_term (lexer "apprend(benjamin, informatique)");;
```

```
[9]: val tokenlist_to_term : token list -> term = <fun>
```

```
[9]: - : term =
Term (Atom "apprend",
      [Term (Atom "benjamin", []); Term (Atom "informatique", [])])
```

### 1.3.5 Transformer les symboles en clauses, le programme en liste de clauses

```
[10]: (* Transforme une liste de symboles représentant une clause en la clause elle-même *)
let tokenlist_to_clause tkcl =
let c = extract_rightleft tkcl in let l = fst c and r = snd c
in Clause (tokenlist_to_term l, if r = [] then [] else List.map
  tokenlist_to_term (extract_term_list r));;

(* Transforme la chaîne de caractères du programme en la liste de clauses du
  format voulu *)
let parser str = List.map tokenlist_to_clause (extract_clauses (lexer str)) ;;
```

```
[10]: val tokenlist_to_clause : token list -> clause = <fun>
```

```
[10]: val parser : string -> clause list = <fun>
```

### 1.3.6 Quelques petits parsers pratiques

```
[11]: (* Un parser sur une chaîne représentant un terme *)
let parser_term str = tokenlist_to_term (lexer str);;

(* Un parser sur une chaîne représentant une liste de termes *)
let parser_term_list str = List.map tokenlist_to_term (extract_term_list (lexer
  str));;

(* Un parser sur une chaîne (sans le point) représentant une clause *)
let parser_clause str = tokenlist_to_clause (lexer str);;

parser_term "apprend(benjamin, informatique)";;

parser_term_list "apprend(E,M), enseigne(P,M)";;

parser_clause "etudiant_de(E,P):-apprend(E,M), enseigne(P,M)"
```

```
[11]: val parser_term : string -> term = <fun>
```

```
[11]: val parser_term_list : string -> term list = <fun>
```

```
[11]: val parser_clause : string -> clause = <fun>
```



```
[11]: - : term =
      Term (Atom "apprend",
            [Term (Atom "benjamin", []); Term (Atom "informatique", [])])

[11]: - : term list =
      [Term (Atom "apprend", [Var (Id ("E", 0)); Var (Id ("M", 0))]);
       Term (Atom "enseigne", [Var (Id ("P", 0)); Var (Id ("M", 0))])]

[11]: - : clause =
      Clause (Term (Atom "etudiant_de", [Var (Id ("E", 0)); Var (Id ("P", 0))]),
              [Term (Atom "apprend", [Var (Id ("E", 0)); Var (Id ("M", 0))]);
               Term (Atom "enseigne", [Var (Id ("P", 0)); Var (Id ("M", 0))])])
```

### 1.3.7 Un exemple

```
[12]: let world = parser program;;
```

```
[12]: val world : clause list =
      [Clause
        (Term (Atom "apprend",
              [Term (Atom "eve", []); Term (Atom "mathematiques", [])]),
         []);
       Clause
        (Term (Atom "apprend",
              [Term (Atom "benjamin", []); Term (Atom "informatique", [])]),
         []);
       Clause
        (Term (Atom "apprend",
              [Term (Atom "benjamin", []); Term (Atom "physique", [])]),
         []);
       Clause
        (Term (Atom "enseigne",
              [Term (Atom "alice", []); Term (Atom "physique", [])]),
         []);
       Clause
        (Term (Atom "enseigne",
              [Term (Atom "pierre", []); Term (Atom "mathematiques", [])]),
         []);
       Clause
        (Term (Atom "enseigne",
              [Term (Atom "pierre", []); Term (Atom "informatique", [])]),
         []);
       Clause (Term (Atom "etudiant_de", [Var (Id ("E", 0)); Var (Id ("P", 0))]),
               [Term (Atom "apprend", [Var (Id ("E", 0)); Var (Id ("M", 0))]);
```

```
Term (Atom "enseigne", [Var (Id ("P", 0)); Var (Id ("M", 0))]);
Clause (Term (Atom "egal", [Var (Id ("R", 0)); Var (Id ("R", 0))]), [])]
```

### 1.3.8 Remarques sur le parser

Je l'ai écrit moi-même, sans reprendre une structure existante... C'est très inefficace : le programme est parcouru au moins 3 fois par les 3 fonctions d'isolement, avant même d'entrer dans la fonction récursive `tokenlist_to_term` qui sert réellement à créer l'arbre... De plus, le programme est copié plein de fois dans les fonctions de parcours. Néanmoins le coût total me paraît de loin proportionnel à la longueur du code, et probablement proportionnel au niveau d'imbrication `((()))` des parenthèses (ce qui pourrait certainement être évité). Je pense que ce n'est pas le plus important : les coûts d'exécution seront bien plus problématiques...

Je pense qu'il pourrait exister encore des codes Prolog dont la syntaxe est fautive et qui passent le parser. (En tous cas je ne suis pas assuré du contraire). Il faut se demander si l'on peut représenter des programmes faux dans les structures formées des types `clause`, `term`, `atom` et `var`.

Remarque : vous êtes priés de ne rentrer que des codes justes dans le parser, les messages d'erreur sont plus que lacunaires...

### 1.3.9 Quelques fonctions d'affichage pour le débogage

```
[13]: let rec term_to_string = function
| Var (Id (str,n)) -> str ^ (string_of_int n)
| Term (Atom a, []) -> a
| Term (Atom a, l) -> a ^ "(" ^ (String.concat ", " (List.map term_to_string ↪
↪ l)) ^ ")";;

let term_list_to_string l = String.concat ", " (List.map term_to_string l);;

let clause_to_string (Clause (h, l)) = (term_to_string h) ^ ":-" ^ ↪
↪ (term_list_to_string l);;
```

```
[13]: val term_to_string : term -> string = <fun>
```

```
[13]: val term_list_to_string : term list -> string = <fun>
```

```
[13]: val clause_to_string : clause -> string = <fun>
```

## 1.4 L'algorithme derrière Prolog

Ma référence en ce qui concerne l'algorithme utilisé par Prolog est *Logic Programming and Prolog*, 1990, Nilsson et Małuszyński [3].

Ce qui semble être le premier article traitant du type de résolution utilisé par Prolog est *A Machine-Oriented Logic Based on the Resolution Principle*, Robinson, 1965 [5]. Cet article introduit l'unification, et le *principe de résolution* (Resolution Principle) dont dérive la SLD-resolution (Linear resolution for Definite clauses with Selection function) utilisée par Prolog.

Une *clause définie*, aussi nommée clause de Horn est de la forme  $(P1 \text{ et } P2 \text{ et } \dots \text{ et } Pn) \Rightarrow Q$ . En particulier,  $(\text{non } P) \Rightarrow Q$  n'est pas une clause de Horn. On peut par contre exprimer  $(P1 \text{ ou } P2) \Rightarrow Q$  avec des clauses de Horn, il suffit d'écrire les deux clauses  $P1 \Rightarrow Q$  et  $P2 \Rightarrow Q$ . Les versions premières de Prolog se limitent aux clauses de Horn, pour une principale raison : la *correction* et la *complétude* (*soundness* and *completeness*) ont été montrée pour la SLD-resolution, qui n'est valide que sur des clauses de Horn. Pour les citations (un peu compliquées) : la SLD-resolution est introduite par Kowalski en 1974, la correction est montrée par Clark en 1979, la complétude a été montrée premièrement par Hill en 1974, mais quelque chose de plus fort a été montré par Clark en 1979.

### 1.4.1 L'unification

L'unification se fait entre deux termes A et B. Deux termes s'unifient s'il existe une substitution  $\theta$  des variables de A telle que  $B = \theta(A)$ .

Soit la clause du programme Prolog `etudiant_de(E,P):-apprend(E,M), enseigne(P,M)`.. Si l'on cherche à réaliser la requête `?-etudiant_de(E, pierre)`, on unifie `etudiant_de(E, pierre)` avec `etudiant_de(E,P)`, le membre de gauche de la clause. La substitution  $\theta$  remplace P par `pierre` et ne modifie pas les autres variables. Pour continuer la recherche, on applique la substitution aux termes de droite de la clause.

Voici l'algorithme décrit dans [3] :

E est l'ensemble des équations. Au départ il n'y en a qu'une seule :  
Par exemple  $E = \{\text{etudiant\_de}(E,P) = \text{etudiant\_de}(E, \text{pierre})\}$ .

Répéter tant que E change

(jusqu'à ce que l'on ne puisse plus rien appliquer aux équations)

Sélectionner une équation  $s = t$  dans E;

Si  $s = t$  est de la forme :

$f(s_1, \dots, s_n) = f(t_1, \dots, t_n)$  avec  $n \geq 0$

Alors remplacer l'équation par  $s_1=t_1 \dots s_n=t_n$

$f(s_1, \dots, s_m) = g(t_1, \dots, t_n)$  avec  $f \neq g$

Alors ÉCHEC

$X = X$

Alors supprimer l'équation

$t = X$  où t n'est pas une variable

Alors remplacer l'équation par  $X = t$

$X = t$  où  $X \neq t$  et X apparaît plus d'une fois dans E

Si X est un sous-terme de t Alors ÉCHEC

Sinon on remplace toutes les autres occurrences de X par t

Il est prouvé que cet algorithme termine et renvoie soit échec, soit un ensemble équivalent des équations sous *forme résolue*. Un ensemble d'équations  $\{X_1 = t_1, \dots, X_n = t_n\}$  est dit sous forme résolue si les  $(X_n)$  sont des variables, les  $(t_n)$  sont des termes et aucune des variables  $X_n$

n'apparaît dans les `tn`. Cela permet ensuite de déterminer un unifieur.

Remarques :

- C'est dommage que l'algorithme se prête si bien à la programmation impérative quand on programme dans un langage fonctionnel.
- On trouve des algorithmes pour déterminer un unifieur qui ne manipulent pas exactement comme ça des systèmes d'équations, (et qui sont dans des styles plus fonctionnels), mais cet algorithme est le seul que j'arrive à comprendre convenablement, et en plus [3] prouve sa terminaison et sa correction, ce qui est très bon point.
- Le test `Si X est un sous-terme de t Alors ÉCHEC` est en pratique pas réalisé dans la plupart des implémentations Prolog (pas comme ça en tout cas), il est très lent et le cas n'arrive que très peu en pratique. On peut donc avoir des boucles infinies... Les versions "modernes" gèrent les unifications de structures infinies... Une citation de [4] "*This represents a circular, infinite unification. Some versions of Prolog, notably Prolog II (Giannesini et al. 1986), provide an interpretation for such structures, but it is tricky to define the semantics of infinite structures.*"

```
[14]: (* Teste si une variable est dans un terme *)
let rec var_in_term v = function
| Var w -> v = w
| Term (_, l) -> List.mem true (List.map (var_in_term v) l);;

var_in_term (Id ("P", 0)) (parser_term "etudiant_de(E,P)");;

(* Remplace une variable par un nouveau terme, récursivement à l'intérieur d'un
↳ terme *)
let rec replace_var_in_term var new_t = function
| Var v when v = var -> new_t
| Var v -> Var v
| Term (a, l) -> Term (a, List.map (replace_var_in_term var new_t) l);;

replace_var_in_term (Id ("E",0)) (parser_term "pierre") (parser_term
↳ "etudiant_de(E,P)");;
```

```
[14]: val var_in_term : var -> term -> bool = <fun>
```

```
[14]: - : bool = true
```

```
[14]: val replace_var_in_term : var -> term -> term -> term = <fun>
```

```
[14]: - : term =
Term (Atom "etudiant_de", [Term (Atom "pierre", []); Var (Id ("P", 0))])
```

```
[15]: (* Teste si une variable est dans une liste de couples de termes *)
let var_in_eql v e = List.mem true (List.map (function t -> var_in_term v (fst
  ↪t) || var_in_term v (snd t)) e);;

var_in_eql (Id ("F", 12)) [Var (Id ("F", 12)), Var (Id ("A", 0)); Var (Id ("A",
  ↪0)), Var (Id ("E", 27))];;

(* Remplace une variable par un nouveau terme, à l'intérieur d'une liste de
  ↪couples de termes *)
let replace_var_in_eql var new_t eql =
List.map (function (a, b) -> (replace_var_in_term var new_t a),
  ↪(replace_var_in_term var new_t b)) eql;;
```

```
[15]: val var_in_eql : var -> (term * term) list -> bool = <fun>
```

```
[15]: - : bool = true
```

```
[15]: val replace_var_in_eql :
  var -> term -> (term * term) list -> (term * term) list = <fun>
```

```
[16]: (* E est une liste de couples de termes *)
(* transforme E en un équivalent sous forme résolue (type option, None si
  ↪impossible) *)
let rec solve e =
  (* Parcourt une fois l'ensemble des équations *)
  let rec pass tmp = function
    | [] -> Some tmp

    | (Term (f, lf), Term (g, lg))::l when f = g && List.length lf = List.
      ↪length lg ->
      pass ((List.map2 (fun a b -> a,b) lf lg)@tmp) l

    | (Term (_, _), Term (_, _))::l -> None (* failwith "Unification failed" *)

    | (Var x, Var y)::l when x = y -> pass tmp l

    | (Term (a, la), Var x)::l -> pass ((Var x, Term (a, la))::tmp) l

    | (Var x, t)::l when var_in_eql x tmp || var_in_eql x l ->
      if var_in_term x t then None (* failwith "Unification failed, loop" *)
      else pass ((Var x, t)::(replace_var_in_eql x t tmp))
      ↪(replace_var_in_eql x t l)

    | (Var x, t)::l -> pass ((Var x, t)::tmp) l
```

```

in match (pass [] e) with
| None -> None
| Some eqlpp -> let b = List.sort_uniq compare eqlpp in if e = b then Some e
  ↪ else solve b;;

```

[16]: `val solve : (term * term) list -> (term * term) list option = <fun>`

À la place de manipuler des ensembles (`Set` existe en OCaml), j’ai juste utilisé des liste triées. La fonction `compare` est magique, elle définit une relation d’ordre pour n’importe quel type. J’ai ici utilisé la version complète de l’algorithme, qui n’est certainement pas la plus efficace. Comme l’a dit Donald Knuth : “*Premature optimization is the root of all evil (or at least most of it) in programming.*”

[17]: `solve [parser_term "etudiant_de(E, P)", parser_term "etudiant_de(S, P)"];; (* E ↪ S *)`

[17]: `- : (term * term) list option = Some [(Var (Id ("E", 0)), Var (Id ("S", 0)))]`

On définit une substitution comme une liste de couples (var,term), le terme est inséré à la place de la variable.

Si l’on sait que `etudiant_de(E1,P) :- apprend(E1,M), enseigne(P,M). (c)` et que l’on veut montrer `etudiant_de(E0, pierre). (r)`, alors il faut trouver une substitution qui unifie la tête de la clause (c) avec la requête (r).

Pour l’exemple plus haut, il faut la substitution `thêta = { P/pierre, E1/E0}`. Il faut comprendre “La variable P est remplacée par ‘pierre’, la variable E1 est remplacée par la variable E0. C’est bien un unifieur, si on l’applique sur la requête et sur la tête de la clause, les deux deviennent égales.

Il peut exister plusieurs unifieurs, par exemple `thêta2 = { P/pierre, E0/samuel, E1/samuel}` en est aussi un. On voit que `thêta2`, c’est `thêta` composé avec `{E0/samuel}`, on dit alors que `thêta` est plus général que `thêta2`. Nous, nous cherchons le plus général de ces unifieurs, le MGU pour *Most General Unifier*.

Un détail : on peut avoir `thêta` plus général que `oméga`, et `oméga` plus général que `thêta`, la relation n’est donc pas antisymétrique. En fait, le MGU est unique au renommage des variables près.

On peut déduire facilement un MGU du système d’équations sous forme résolue de la fonction `solve` (c’est à ça qu’elle sert). Si `{X1 = t1, ..., Xn = tn}` est sous forme résolue, alors `{X1/t1, ..., Xn/tn}` est un MGU.

Lorsque l’on veut unifier `etudiant_de(E,P)` et `etudiant_de(E,pierre)`, les deux variables E ne doivent pas être nommées de la même manière.

[18]: `(* Prend 2 terme et renvoie l'unifieur le plus général s'il existe, None sinon. Attention, aucune variable ne doit être présente dans (r) la requête et dans ↪ (c) la tête de la clause *)`

```

let rec mgu r c = let f = function
  | Var x, Term (a, l) -> x, Term (a, l)
  | Var x, Var y -> x, Var y
  | Term (_, _), _ -> failwith "Should not happen."
in match solve [r, c] with
| None -> None
| Some e -> Some (List.map f e);;

```

[18]: val mgu : term -> term -> (var \* term) list option = <fun>

```

[19]: (* La fonction test_unification a pour but de tester mgu *)
let test_unification str_r str_c = mgu (parser_term str_r) (parser_term str_c);;

test_unification "etudiant_de(E, pierre)" "etudiant_de(F,P)";; (* F/E et P/
    ↪ pierre *)

test_unification "etudiant_de(F,P)" "etudiant_de(E, pierre)";; (* E/F et P/
    ↪ pierre *)

test_unification "f(X,g(Y))" "f(g(Z),Z)";; (* X/g(g(Y)) et Z/g(Y) *)

test_unification "f(g(Z),Z)" "f(X,g(Y))";; (* X/g(g(Y)) et Z/g(Y) *)

```

[19]: val test\_unification : string -> string -> (var \* term) list option = <fun>

[19]: - : (var \* term) list option =  
Some  
[(Id ("E", 0), Var (Id ("F", 0))); (Id ("P", 0), Term (Atom "pierre", []))]

[19]: - : (var \* term) list option =  
Some  
[(Id ("F", 0), Var (Id ("E", 0))); (Id ("P", 0), Term (Atom "pierre", []))]

[19]: - : (var \* term) list option =  
Some  
[(Id ("X", 0), Term (Atom "g", [Term (Atom "g", [Var (Id ("Y", 0))]))]);  
(Id ("Z", 0), Term (Atom "g", [Var (Id ("Y", 0))]))]

[19]: - : (var \* term) list option =  
Some  
[(Id ("X", 0), Term (Atom "g", [Term (Atom "g", [Var (Id ("Y", 0))]))]);  
(Id ("Z", 0), Term (Atom "g", [Var (Id ("Y", 0))]))]

Quelques explications sur les derniers deux exemples :

- Je sais que pour tout  $Z$ ,  $f(g(Z), Z)$  est vraie, et j'aimerais savoir s'il existe des couples  $(X, Y)$  tels que  $f(X, g(Y))$  soit vraie. Le programme me répond oui, il suffit de choisir  $X = g(g(Y))$
- Je sais que pour tout  $(X, Y)$ ,  $f(X, g(Y))$  est vraie, et j'aimerais savoir s'il existe des  $Z$  tels que  $f(g(Z), Z)$  soit vraie. Le programme me répond oui, il suffit de choisir  $Z = g(Y)$

### 1.4.2 Le backtracking

Il me semble que nous nous rapprochons du but ! L'unification est une partie très importante de la SLD-resolution. L'autre point important est le backtracking. Encore quelques fonctions pour appliquer des substitutions sur des termes, listes de termes, pour composer des substitutions, pour vérifier que un terme et une clause n'ont pas de variables en commun, pour effectuer les renommages si nécessaire, et après je pense qu'il sera possible d'implémenter le backtracking.

```
[20]: (* Applique une substitution sur un terme *)
let rec apply_subst_on_term uni term = match uni with
| [] -> term
| (v,t)::l -> apply_subst_on_term l (replace_var_in_term v t term);;

(* Applique une substitution sur une liste de termes *)
let rec apply_subst_on_termlist uni lst = List.map (apply_subst_on_term uni) lst;
  ↪lst;;
```

```
[20]: val apply_subst_on_term : (var * term) list -> term -> term = <fun>
```

```
[20]: val apply_subst_on_termlist : (var * term) list -> term list -> term list =
  <fun>
```

Voilà la manière dont j'ai compris l'algorithme :

- On cherche à satisfaire une requête  $\leftarrow A_1, \dots, A_n$ .
- On a  $A_1$  qui unifie avec  $H_i$ , où  $H_i \leftarrow C_{i_1}, \dots, C_{i_m}$  est une la  $i$ -ème clause du programme. On a auparavant renommé toutes les variables de  $H_i \leftarrow C_{i_1}, \dots, C_{i_m}$  qui étaient présentes dans  $A_1$ , sinon on ne peut pas appeler mgu. On a donc un MGU  $\theta_{i_1}$ .
- La nouvelle requête à satisfaire est  $\leftarrow \theta_{i_1}(C_{i_1}, \dots, C_{i_m}, A_2, \dots, A_n)$ .
- On récurse, si on a à montrer  $\theta_k( \dots \theta_{i_2}( \theta_{i_1}( \_vide\_ ) ) \dots )$  alors c'est gagné ( $\_vide\_$  signifie toujours vrai), la composée  $\theta_{tot}$  des  $\theta_k$  est une substitution des variables de la requête de départ. Ce qu'on veut renvoyer à l'utilisateur c'est l'image des variables de  $A_1, \dots, A_n$  par  $\theta_{tot}$ .
- Si  $A_1$  ne s'unifie avec aucun  $H_i$ , ça ne sert à rien d'essayer d'unifier  $A_2$ , puisque de toute façon on garde  $A_1$  dans la nouvelle requête. Alors il faut remonter. C'est à dire qu'il faut essayer les autres unifications à l'étape d'avant.



Comment renommer les clauses pour avoir des noms libres à chaque unification ? On peut utiliser le niveau de récursion comme identifiant, que l'on place dans l'entier transporté avec la variable.

- On veut des variables numérotées à 0 dans la requête.
- On passe 1 lors du premier appel de sld : la première clause utilisée est renommée à 1, les variables de la requête sont toutes à 0.
- La deuxième clause utilisée est renommée à 2, dans la requête il y a des variables à 1 et à 0 : pas de conflit ! etc...

```
[21]: (* Prend un entier et une clause, renvoie la clause avec les variables ↪
↪renommées à l'entier *)
let rename n (Clause (t1, t1)) =
let rec f = function
| Var (Id (str, _)) -> Var (Id (str, n))
| Term (atm, l) -> Term (atm, List.map f l)
in Clause (f t1, List.map f t1);;

rename 42 (parser_clause "etudiant_de(E,P):-apprend(E,M), enseigne(P,M)");;
```

```
[21]: val rename : int -> clause -> clause = <fun>
```

```
[21]: - : clause =
Clause (Term (Atom "etudiant_de", [Var (Id ("E", 42)); Var (Id ("P", 42))]),
[Term (Atom "apprend", [Var (Id ("E", 42)); Var (Id ("M", 42))]);
Term (Atom "enseigne", [Var (Id ("P", 42)); Var (Id ("M", 42))])])
```

```
[22]: (* La fonction de recherche, s'arrête à la première solution trouvée *)
let rec sld (world : clause list) (req : term list) (subs : (var * term) list ↪
↪list) n =
(*Format.printf "%n %s\n%!" n (term_list_to_string req);*)
match req with
| [] -> Some subs
| head_request_term::other_request_terms ->
let rec f = function
| [] -> None
| c::other_clauses -> let Clause (left_member, right_member) = rename n c in
(match mgu head_request_term left_member with
| None -> f other_clauses
| Some unifier -> (*Format.printf "Used %s\n%!" (clause_to_string ↪
↪(rename n c));*)
(match sld world
(apply_subst_on_termlist unifier (right_member@other_request_terms))
(unifier::subs) (n+1) with
| None -> (*Format.printf "Failed\n%!" ;*) f other_clauses
| Some subst -> Some subst )
```

```
)
in f world;;
```

```
[22]: val sld :
      clause list ->
      term list -> (var * term) list list -> int -> (var * term) list list option =
      <fun>
```

### 1.4.3 Déchiffrer les retours de la fonction sld

```
[23]: (* Compose correctement la liste de substitutions *)
let compose l =
let compose2 s1 s2 = s1@(List.map (function (v2, lt2) -> (v2,
  ↳apply_subst_on_term s1 lt2)) s2)
in List.fold_left compose2 [] l;;

(* Recherche les termes qui sont des variables récursivement dans une liste de
  ↳termes *)
let rec find_vars_in_termlist tl = List.sort_uniq compare (List.concat (List.map
(function
| Var v -> [Var v]
| Term (_,l) -> find_vars_in_termlist l)
tl))

(* Effectue une requête *)
let request world req = match (sld world (parser_term_list req) [] 1) with
| None -> None
| Some l -> Some (apply_subst_on_termlist (compose l) (find_vars_in_termlist
  ↳(parser_term_list req)));;
```

```
[23]: val compose : (var * term) list list -> (var * term) list = <fun>
```

```
[23]: val find_vars_in_termlist : term list -> term list = <fun>
```

```
[23]: val request : clause list -> string -> term list option = <fun>
```

### 1.4.4 Enfin des essais !

```
[24]: request world "etudiant_de(E, pierre)";;

request world "etudiant_de(E, pierre), etudiant_de(E, alice)";;

request world "etudiant_de(A, B)";;
```

```
request world "etudiant_de(A, A)";;

request world "apprend(A, A)";;

request world "enseigne(A, A)";;

request world "enseigne(alice, physique)";;

request world "enseigne(alice, mathematiques)";;
```

```
[24]: - : term list option = Some [Term (Atom "eve", [])]
```

```
[24]: - : term list option = Some [Term (Atom "benjamin", [])]
```

```
[24]: - : term list option = Some [Term (Atom "eve", []); Term (Atom "pierre", [])]
```

```
[24]: - : term list option = None
```

```
[24]: - : term list option = None
```

```
[24]: - : term list option = None
```

```
[24]: - : term list option = Some []
```

```
[24]: - : term list option = None
```

## 1.5 Que faire maintenant ?

- Beaucoup de tests pour vérifier si tout fonctionne.
- Peut-être nettoyer un peu le code...
- Écrire quelques fonctions comme une ligne de commande interactive, pour une utilisation plus facile. Rendre les résultats plus clairs.
- Proposer une deuxième version de la recherche, qui renvoie tous les résultats possibles, ou qui propose de chercher le résultat suivant.
- Je pensais essayer d'écrire un programme pour “résoudre” le Cluedo. J’avais déjà essayé sans succès en C++. Le problème semble plus adapté au langage Prolog : “*tel joueur possède telle carte*”, “*si un joueur montre une carte à un autre pour réfuter une supposition (triplet de*

*cartes*), *c'est qu'il possède une des trois cartes*", sont des règles que l'on peut écrire en Prolog. Malheureusement j'ai trouvé un super mémoire de Master [1] qui traite bien le sujet.

- Je regarde un peu comment est-ce que l'on peut sortir des clauses de Horn, mais c'est un sujet compliqué.
- Je voulais voir s'il est possible d'étendre à des langages fonctionnels des éléments de programmation logique. C'est aussi un thème intéressant, qui se nomme *functional logic programming*. On trouve plusieurs petits langages sur internet qui utilisent des approches différentes, mais aucun d'eux ne semble bien abouti. Ce serait presque un idéal d'associer des mécanismes efficaces des langages fonctionnels comme l'évaluation paresseuse, et les capacités de la programmation logique pour décrire les problèmes...
- Pour compiler Prolog, il faut se tourner vers les *Machines abstraites de Warren*.

## Références

- [1] V. I. Aartun. Reasoning about knowledge and action in cluedo using prolog. 2016.
- [2] P. Flach. *Simply Logical : Intelligent Reasoning by Example*. 1994.
- [3] U. Nilsson and J. Małuszyński. *Logic, Programming and Prolog*. 1990.
- [4] P. Norvig. *Paradigms of Artificial Intelligence Programming*. 1992.
- [5] J. A. Robinson. A machine-oriented logic based on the resolution principle. 1965.
- [6] A. Smaill. Logic programming. 2009.
- [7] P. Weis and X. Leroy. *Le langage Caml*. 1992.